Correction du partiel du 4 novembre

Remarque générale : dans les copies, un certain nombre d'erreurs sont dûes à une mauvaise lecture de l'énoncé. Lisez attentivement la question avant d'y répondre ! Si on demande une valeur approchée, ne calculez pas une valeur exacte. Si on demande de définir une fonction, ne définissez pas une expression. Si on demande d'utiliser *D*, n'utilisez pas *diff* ! Etc...

Exercice 1

1)

```
> restart;
  with(numtheory);
Warning, the protected name order has been redefined and unprotected
```

[*GIgcd*, *bigomega*, *cfrac*, *cfracpol*, *cyclotomic*, *divisors*, *factorEQ*, *factorset*, *fermat*, *imagunit*, *index*, *integral_basis*, *invcfrac*, *invphi*, *issqrfree*, *jacobi*, *kronecker*, λ, *legendre*, *mcombine*, *mersenne*, *migcdex*, *minkowski*, *mipolys*, *mlog*, *mobius*, *mroot*, *msqrt*, *nearestp*, *nthconver*, *nthdenom*, *nthnumer*, *nthpow*, *order*, *pdexpand*, ϕ, π, *pprimroot*, *primroot*, *quadres*, *rootsunity*, *safeprime*, σ, *sq2factor*, *sum2sqr*, τ, *thue*]

```
> ?numtheory
```

La commande de la librairie *numtheory* qui permet de calculer les diviseurs d'un nombre entier est *divisors* (pour consulter son aide, taper *?numtheory[divisors]*)

```
> n:=4568742320;
```

$$n := 4568742320$$

```
> divisors(n);
```

{1, 2, 4, 5, 8, 10, 16, 20, 40, 80, 193, 386, 772, 965, 1544, 1930, 3088, 3860, 7720, 15440, 295903, 591806, 1183612, 1479515, 2367224, 2959030, 4734448, 5918060, 4568742320, 1142185580, 2284371160, 11836120, 23672240, 57109279, 114218558, 228437116, 285546395, 456874232, 571092790, 913748464}

2) La commande précédente nous a donné un *ensemble* de diviseurs. Pour calculer le nombre d'éléments d'un ensemble (ou d'une liste), on dispose de la commande *nops*.

```
> nops(%);
```

$$40$$

Certains ont remarqué, à juste titre, que la commande *tau* de la librairie permet de calculer le nombre de diviseurs :

```
> tau(n);
```

$$40$$

3) Les diviseurs communs à n et 154400 sont les diviseurs de n qui sont aussi diviseurs de 154400. Pour les déterminer, il suffit de faire l'intersection de l'ensemble des diviseurs de n avec l'ensemble des diviseurs de 154400.

```
> divisors(n) intersect divisors(154400);
```

{1, 2, 4, 5, 8, 10, 16, 20, 40, 80, 193, 386, 772, 965, 1544, 1930, 3088, 3860, 7720, 15440}

Exercice 2

1) On fait une procédure récursive, dans laquelle on fait un test avec *if* pour distinguer les différents cas.

```
> restart;
> u:=proc(n)
  if n<=0 then RETURN("Erreur")
  elif n=1 then RETURN(2)
  else RETURN(u(n-1)/2+1/u(n-1));
  fi;
  end;
```

u := **proc**(*n*)
  if $n \le 0$ **then** RETURN("Erreur")
  elif $n = 1$ **then** RETURN(2)
  else RETURN(1 / 2∗u($n-1$) + 1 / u($n-1$))
  end if
end proc

Remarques :
- pour afficher une chaîne de caractères, on utilise " "
- **la formule de récurrence donnée exprime u(n+1) en fonction de u(n). Pour définir la suite dans Maple, on a besoin de la ramener à la forme : u(n) en fonction de u(n-1).**

2) On utilise *seq*.

```
> seq(evalf(u(k)),k=1..10);
```

2., 1.500000000, 1.416666667, 1.414215686, 1.414213562, 1.414213562, 1.414213562, 1.414213562, 1.414213562, 1.414213562

```
> evalf(sqrt(2));
```

$$1.414213562$$

On constate que la suite semble converger vers $\sqrt{2}$.

Exercice 3

1) On nous demande de définir une *fonction*.

```
> restart;
> f:=x->((x^2-x-6)^2-x+3)/(x^2-4*x+3);
```

$$f := x \to \frac{(x^2-x-6)^2-x+3}{x^2-4\,x+3}$$

```
> discont(f(x),x);
```

$$\{1, 3\}$$

Donc f est continue sur Df=]-∞,1[U]1,3[U [3, ∞[.

```
> iscont(f(x),x=-infinity..1);
```

true

```
> iscont(f(x),x=1..3);
```

true

```
> iscont(f(x),x=3..infinity);
```

true

2)

```
> limit(f(x),x=-infinity);
```

$$\infty$$

```
> limit(f(x),x=infinity);
```

$$\infty$$

```
> limit(f(x),x=1);
```

$$undefined$$

```
> limit(f(x),x=1,left);
```

$$\infty$$

```
> limit(f(x),x=1,right);
```

$$-\infty$$

```
> limit(f(x),x=3);
```

$$\frac{-1}{2}$$

Comme la limite de $f(x)$ en 1 vaut ∞ à gauche et $-\infty$ à droite, la courbe C admet une asymptote verticale d'équation $x=1$.

3) Comme la limite de $f(x)$ en 3 est finie et vaut $-\frac{1}{2}$, la fonction f peut être prolongée par continuité en $x=3$ en posant $f(3)=-\frac{1}{2}$.

```
> simplify(f(x));
```

$$\frac{x^3+x^2-8x-13}{x-1}$$

Sur l'expression simplifiée, on voit que 3 n'apparaît plus comme zéro du dénominateur. Cette expression est continue en 3. On retrouve le fait que f est prolongeable par continuité en 3.

4) Pour calculer la dérivée avec D :

```
> g:=D(f);
```

$$g := x \to \frac{2(x^2-x-6)(2x-1)-1}{x^2-4x+3} - \frac{((x^2-x-6)^2-x+3)(2x-4)}{(x^2-4x+3)^2}$$

Remarque : on peut lui donner meilleure allure en mettant tout au même dénominateur :

```
> normal(g(x));
```

$$\frac{2x^3-2x^2-2x+21}{(x-1)^2}$$

Calculer les racines de g.

```
> S:=solve(g(x),x);
```

$$S := -\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} - \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3}, \frac{(1090+6\sqrt{32889})^{(1/3)}}{12} + \frac{4}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3} + \frac{1}{2} I\sqrt{3}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} + \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}\right), \frac{(1090+6\sqrt{32889})^{(1/3)}}{12} + \frac{4}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3} - \frac{1}{2} I\sqrt{3}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} + \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}\right)$$

```
> evalf(S);
```

$$-2.032828057,\ 1.516414027 - 1.692839762\,I,\ 1.516414027 + 1.692839762\,I$$

On voit qu'il y a une racine réelle (la première) et deux racines complexes. On nous demande la racine réelle , donc on ne garde que la première.

```
> a:=S[1];
```

$$a := -\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} - \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3}$$

Etude du signe de g.

```
> solve(g(x)>0,x);
```

$$\mathrm{RealRange}\left(\mathrm{Open}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} - \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3}\right), \mathrm{Open}(1)\right), \mathrm{RealRange}(\mathrm{Open}(1), \infty)$$

```
> solve(g(x)<0,x);
```

$$\mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6} - \frac{8}{3(1090+6\sqrt{32889})^{(1/3)}} + \frac{1}{3}\right)\right)$$

Donc g est strictement négative sur]- ∞,a[, puis strictement positive sur]a,1[U]1, +∞[. On en déduit que f est strictement décroissante sur]- ∞,a[, puis strictement croissante sur]a,1[U]1, +∞ [.

5)

```
> solve(f(x),x);
```

$$\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}, -\frac{(1108+12\sqrt{1581})^{(1/3)}}{12} - \frac{25}{3(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3} + \frac{1}{2} I\sqrt{3}\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3(1108+12\sqrt{1581})^{(1/3)}}\right), -\frac{(1108+12\sqrt{1581})^{(1/3)}}{12} - \frac{25}{3(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3} - \frac{1}{2} I\sqrt{3}\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3(1108+12\sqrt{1581})^{(1/3)}}\right)$$

On voit qu'il y a une racine réelle (la première) et deux racines complexes. On nous demande la racine réelle , donc on ne garde que la première.

```
> z:=solve(f(x),x)[1];
```

$$z := \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}$$

La tangente à *C* en (*z*, 0) a pour équation : y = f(z) + (x-z) f'(z) c'est-à-dire y = (x-z) g(z).

```
> T:=(x-z)*g(z);
```

$$T := \left(x - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{1}{3}\right)\Bigg(\Bigg(2\Bigg(\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\right)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{17}{3}\Bigg)\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{3} + \frac{100}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{5}{3}\right) - 1\Bigg) \Bigg/ \Bigg(\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\right)^2 - \frac{2\,(1108+12\sqrt{1581})^{(1/3)}}{3} - \frac{200}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{13}{3}\Bigg) - \Bigg(\Bigg(\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\right)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{17}{3}\Bigg)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{10}{3}\Bigg)\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{3} + \frac{100}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{14}{3}\right) \Bigg/ \Bigg(\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\right)^2 - \frac{2\,(1108+12\sqrt{1581})^{(1/3)}}{3} - \frac{200}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{13}{3}\Bigg)^2\Bigg)$$

6)

```
> plot([f(x),T],x);
```



On recadre en abscisse et en ordonnée pour voir clairement la courbe et sa tangente. Par exemple ainsi :

```
> plot([f(x),T],x=-5..5,y=-30..30);
```